BUREAU OF MINERAL RESOURCES, GEOLOGY AND GEOPHYSICS

RECORD

RECORD 1987/45

ACCELEROGRAPH TIMING SYSTEM

by

G.W. BLACK

RECORD 1987/45

ACCELEROGRAPH TIMING SYSTEM

by

G.W. BLACK

CONTENTS

ILLUSTRATIONS

FIGURE

PREFACE

In 1983 a request was made from EARTHQUAKE RISK to modify existing MO2 and SMA-1 accelerographs.

The required modifications involved a method of establishing the date and time of a recorded seismic event, and to increase the overall sensitivity of the trigger mechanism, and in particular higher sensitivity in the range 5 to 25 Hz.

This manual covers the operation, maintenance and installation of these modifications.

GENERAL DESCRIPTION

Minor modifications were necessary to the accelerographs to enable the time/date information to be recorded on the film. All the electronics are externally housed in a die-cast box and are connected to the accelerograph via a 19 pin cannon connector.

Time and date information is maintained by a low power OKI I.C. (MSM5832), this unit was chosen for its low power consumption during standby conditions.

A Motorola MC68705U3 microprocessor is used to enable setting of the time and date information, and to control the strobing of this information onto the film.

Sensitivity has been increased by using two LM358 dual low power op-amps to provide overall gain increase of 0-36 db in 6 db steps, and an adjustable gain of +/- 6 db at 25 Hz.

SPECIFICATIONS

OVERALL GAIN ..... 0 TO 36 db.

GAIN STEPS ..... 6db.

HIGH FREQUENCY GAIN ..... +/- 0 db @ 5Hz.
+/- 1.6 db @ 10Hz.
+/- 3.5 db @ 15Hz.
+/- 6 db @ 25Hz.
+/- 10 db @ 50Hz.

HIGH FREQUENCY ROLLOFF ..... 3 db/octave ABOVE 150 Hz.

TIME ACCURACY (ADJUSTABLE) ..... +/- 0.2 SEC PER DAY

TIME/DATE RECORDING ..... BINARY CODED DECIMAL
(ON FILM)

CURRENT CONSUMPTION ..... 3mA IN STANDBY MODE
130mA IN RECORDING OR SET MODE

OPERATION

DISPLAY



The display consists of 12 digits, representing from left to right years, months, day of month, hours, minutes and seconds. The clock operates in 24 hour mode.

Below the display there are three switches and two knobs;

FILM/SET

This toggle switch enables the operator to SET (or check) the time/date information, when completed the switch is left in FILM position to enable normal operation of the time/date recording system.

DIGIT

When in the SET mode and the DIGIT button is pressed, the cursor will move across the display at a one second rate, stopping under each time/date character that can be adjusted by the operator (the seconds cannot be adjusted).

ADVANCE

When in the SET mode and the cursor has been positioned on the display, by pressing the ADVANCE button the character above the cursor will be advanced once per second with no roll over occurring.

By using the switch and push buttons the time and date information can be set to the nearest minute.

NOTE. No checking is applied to the validity of the data. e.g. It is the responsibility of the operator not to set FEB 31st.

OVERALL GAIN

The overall gain can be set from 0 to 36 db in steps of 6 db by moving the seven position rotary switch to the appropriate position.

HIGH FREQUENCY GAIN

A linear potentiometer changes the high frequency gain characteristics. As shown in Fig. 1 fully clockwise will provide maximum gain, whilst fully anticlockwise provides minimum gain (maximum attenuation) and at the mid point of it's adjustment, the gain response is flat.

Neither the overall nor high frequency gain controls have any effect on the data recorded on the film, their only effect is on the trigger mechanism. Thus the sensitivity and damping of the relevant accelerographs remain as original.

To establish the best overall and high frequency gain settings it will be necessary to install the accelerograph at its normal operating site with nominal settings of say, overall gain 12 db and high frequency gain at maximum clockwise. After some events have been recorded, it will be necessary to recover and process the film, then inspect the recorded information to see if the sensitivity of the instrument should be increased or decreased and adjust the gain settings as necessary, until triggering of the instrument occurs at the level required.

If gain in the trigger circuit or time keeping is not required, disconnect the electronics in the die-cast box and fit the DUMMY PLUG and the accelerograph will work as originally manufactured.

FIG. 1


11
10
9
8
7
6
5
4
3
2
1
0
−1
−2
−3
−4
−5
−6
−7
−8
−9
−10
−11
H.F. GAIN Db
0.10
0.20
0.40
0.80
1.60
3.20
6.40
12.80
25.60
51.20
FREQUENCY Hz
MAX CLOCKWISE
50% CLOCKWISE
MID POSITION
50% ANTI-CLOCKWISE
MAX ANTI -CLOCKWISE

DATA RECOVERY

After a film has been recovered and processed, a five second area of B.C.D. data will be evident during the last ten seconds of each film run. The data is recorded in the following manner:

YEARS *10
YEARS *1
MONTHS *10
MONTHS *1
DAYS *10
DAYS *1
HOURS *10
HOURS *1
MINUTES *10
MINUTES *1

and will appear on the film as:



The above code will indicate 27 MAY 86 at 10:06

NOTE The zero appears as a blank section on the film.

CIRCUIT DESCRIPTION AND ADJUSTMENTS

DESCRIPTION

Refer to drawing S18 Sheet 1

STANDBY

Power is supplied from the accelerograph batteries via the cannon connector.

During standby mode A3 is supplied with approx. 3 volts via CR2 and CR3 to maintain time keeping whilst in a low power mode. 12 volts is also supplied to the gain stage consisting of A1 and A2.

The signal from the trigger coil is buffered by one half of A1, configured as a differencing amplifier with a gain of 1, R1 and C2 provide a high frequency roll off to reduce any noise introduced in the input stage.

Amplification of the signal is provided by the second half of A1, configured as inverting amplifier with switch selectable gain between 0 and 36 db, in 6 db steps.

One half of A2 is configured as an inverting amplifier with a gain of 1 for any signal below 5 Hz, above this frequency the feedback circuit, consisting of C4, R17, R18, R19, and C7 modifies the gain of this stage, such that a gain variation +/- 6 db can be achieved at 25Hz, depending upon the setting of R18.

The output of this stage is A.C. coupled by C9 to provide the trigger for MO2 accelerographs.

An inverting amplifier with a gain of 1 provides the necessary differential signal via coupling capacitor C8 to trigger the SMA-1 accelerograph.

Connections shown in FIG. 12 illustrate how the different input stages of the SMA-1 and MO2 accelerographs are connected to the gain stages.

RECORDING

When an event is being recorded, a 12 volt signal is applied to the base of transistor switch Q1, this then supplies power to all the stages which only work when time/date information is required.

Voltage regulator A6 provides 5 volts for A3, A5 and DS1.

Output of the sense line from the accelerograph will be near 12 volt (at turn on) and will decay via an R/C circuit (in the accelerograph), at approx 3-4 volts the accelerograph will turn off.

Comparator A4 compares the output of the sense line against the reference voltage from R33, when the output of A4 toggles from 0 to 5 volts, the C.P.U., A5 will detect this and then write the time/date information to the film. R33 is adjusted to allow the time/date information to be written approx 10 sec prior to the accelerograph switching off.

All time setting and writing of data to the film is handled under the control of A5 the microprocessor. Self clocking of A5 at approx. 1MHz. is achieved by R35. C19 provides the necessary time constant for the C.P.U. reset routine.

Timekeeping accuracy can be adjusted to +/- 0.2 seconds per day by C13.

Display DS1, a 16 character L.C.D. is not enabled when in the recording mode.

TIME SET

Switch S3B bypasses Q1 to supply power to the C.P.U. display and clock, at the same time, changing the voltage on pin 22 of A5. The software detects this voltage on pin 22 and enables the time set mode.

S2 and S1 may now be activated to alter the time/date information which is displayed on DS1. R32 provides an adjustment of the viewing angle of the L.C.D.

ADJUSTMENTS

DELAY

R33 adjusts the point on the film that the time/date information is written, normally this would be set to occur approx. 10 seconds before the film stops motion.

To check R33 adjustment, it is necessary to dismantle the accelerograph so that the illumination of the L.E.D.s can be observed. Once in this configuration, with the FILM/SET switch in the FILM position, trigger the accelerograph. After a short period of time the L.E.D.s can be seen flashing, R33 can be adjusted so that the delay from the start of the L.E.D.s flashing to when the accelerograph stops recording is approx. 10 seconds.

If an adjustment is necessary it will be necessary to alter R33,

retrigger the accelerograph and time the delay, then repeat as required.

VIEWING ANGLE

R32 varies the viewing angle of the L.C.D.

To alter set the FILM/SET switch to SET and adjust R32 so that a satisfactory viewing angle is obtained.

CLOCK RATE

C13 is used to adjust the crystal of A3 to a frequency of 32.768KHz.

With the FILM/SET switch in SET position the voltage on pin 10 of A3 is 5 volt, this voltage will change to 0 volt and then toggle between 0 and 5 volts for approx. 8 milli seconds finally returning to 5 volts, this sequence reoccurs every 1.000000 seconds.

The only accurate method to adjust C13 is to measure the delay between negative transitions at pin 10 of A3 with a time interval delay of >20 milli seconds and <900 milli seconds, and adjust C13 to obtain a reading of 1.000000 seconds.

L.E.D. INTENSITY

Resistors 28,29,30 and 31 limit the current to the L.E.D.s which record the time/date information on the film.

To vary the current (and the density) it is necessary to replace these resistors with different values.

MODIFICATIONS TO SMA-1 ACCELEROGRAPHS

Dismantle sufficient of the film transport mechanism to remove the central bar behind the film sprocket.

Drill four holes as per FIG. 6 to enable installation of four miniature L.E.D.s.

Glue the L.E.D.s in the drilled holes so that the cathodes are on the top (or bottom).

Mount a 19 pin cannon connector on the outside of the accelerograph case, ensuring that the mounting is light and water tight.

Connect the internal wiring loom as per the list in FIG. 5, ensure that the cables from the trigger coil are disconnected from their existing points, routed to the connector and back to their original points.

Note the 'sense' connection to the original circuit as shown in FIG. 2.

Reassemble the accelerograph, connect the external electronics and power.

The unit is now ready to test and adjust.

MODIFICATIONS TO MO2 ACCELEROGRAPHS

Remove the light post which houses the existing lamp for timing marks.

Drill four holes as per FIG. 7 to enable installation of four miniature L.E.D.s.

Glue the L.E.D.s in the drilled holes so that the cathodes are on the left (or right).

Mount a 19 pin cannon connector on the outside of the accelerograph case immediately to the left of the existing connector, ensuring that the mounting is light and water tight.

It is necessary to remove the bottom panel of the accelerograph housing to facilitate wiring connections.

Connect the internal wiring loom as per the list in FIG. 4 and the following guide;

12 volt	pin 4	edge	connector
12 volt switched	pin 1	"	"
ground	pin 5	"	"
sense	pin13	"	"

ensure that the cables from the trigger coil are disconnected from their existing points, routed to the connector and back to their original points.

To enable the sense connection to be made as shown in FIG. 3, a wire needs to be added to the printed circuit board as shown in the photograph in FIG. 8.

Reassemble the accelerograph, connect the external electronics and power.

The unit is now ready to test and adjust.



FIG. 2



FIG. 3

Bureau of Mineral Resources, Geology and Geophysics, P.O. Box 378, Canberra City, A.C.T. 2601	**Drawing No.**	**Issue**	**Title:** SMA-1, M02 STARTER CIRCUIT CONNECTIONS
	Sheet No.		
	Drawn G.W.B **Date** 27·6·86		

MD 2 INTERNAL CONNECTIONS

CONNECTOR PINS.



F N.C.

G +12V RED

H +12V SWITCHED. ORANGE

J N.C.

K GROUND GREEN

L N.C.

M FROM TRIGGER COIL. BROWN

N N.C.

P TO INT. TRIGGER GREEN

R N.C.

S FROM TRIGGER COIL. BLUE

T TO INT. TRIGGER (GND.) BLACK

U N.C.

V SENSE BLACK

M, P, S, T — DUMMY PLUGS.

FIG 4

ACCELEROGRAPH TIMING SYSTEM

Bureau of Mineral Resources, Geology and Geophysics, P.O. Box 378, Canberra City, A.C.T. 2601	Drawing No. SIB	Issue 1	Title: MD 2 INT. CONNECTIONS
	Sheet No.		
	Drawn RGal Date 3/6/86		

SMA INTERNAL CONNECTIONS



FIG 5

ACCELEROGRAPH TIMING SYSTEM

Bureau of Mineral Resources, Geology and Geophysics, P.O. Box 378, Canberra City, A.C.T. 2601	Drawing No. S1B	Issue 1	Title: SMA INT. CONNECTIONS.
	Sheet No.		
	Drawn [signature] Date 3/6/86		





FIG 6

SMA_1

Bureau of Mineral Resources, Geology and Geophysics, P.O. Box 378, Canberra City, A.C.T. 2601	Drawing No. S18	Issue 1	Title: LED MOUNTING DRILLING DETAILS
	Sheet No. 5		
	Drawn [signature] Date 22/6/86		



NOTE:-

UNLESS OTHERWISE SHOWN, ALL DIMENSIONS IN MILLIMETRES.

FIG 7

ACCELEROGRAPH TIMING SYSTEM

Bureau of Mineral Resources, Geology and Geophysics, P.O. Box 378, Canberra City, A.C.T. 2601	Drawing No. S18	Issue 1	Title: LED MTG. FOR M02 DRILLING DETAILS
	Sheet No.		
	Drawn [signature] Date 11/6/86		



FIG. 8

M02 P.C.B. MODIFICATIONS

CLOCK CONNECTIONS



FIG 9

ACCELEROGRAPH TIMING SYSTEM

Bureau of Mineral Resources, Geology and Geophysics, P.O. Box 378, Canberra City, A.C.T. 2601	Drawing No. 518	Issue 1	Title: CLOCK CONNECTIONS
	Sheet No.		
	Drawn [signature] Date 4/6/86		



FIG 10

ACCELEROGRAPH TIMING SYSTEM

Bureau of Mineral Resources, Geology and Geophysics, P.O. Box 378, Canberra City, A.C.T. 2601	Drawing No. S18	Issue	Title: FRONT PANEL DRILLING DETAILS
	Sheet No.		
	Drawn Date		



FIG 11

Bureau of Mineral Resources, Geology and Geophysics, P.O. Box 378, Canberra City, A.C.T. 2601	Drawing No.	Issue	Title: FRONT PANEL LABEL TEMPLATE
	Sheet No.		
	Drawn G.W.B Date 19-6-86		



MO2



DUMMY PLUG

SMA-1





DUMMY PLUG

FIG. 12

Bureau of Mineral Resources, Geology and Geophysics, P.O. Box 378, Canberra City, A.C.T. 2601	**Drawing No.**	**Issue**	**Title:** AMPLIFIER CONNECTIONS
	Sheet No.		
	Drawn G.W.B **Date** 27·6·86		


SENS
IN
GND
SMA TRIG
MO2 TRIG
SOB 17
12V IN
COM
VS+
DS WR
GND
INP.
12V

FIG 13 P.C.B

ITEM	VALUE	Vw	%	TYPE	MAKE	REF	BMR VOCAB	QTY
A1-2				DUAL OP-AMP	NAT.	LM358	N.I.V.	2
A3				COMPUTER CLOCK	OK1	MSM-5832	N.I.V.	1
A4				VOLTAGE COMPARATOR	NAT.	LM311	BUA-LM311H	1
A5				C.P.U.	MOT.	MC687-05U3	N.I.V.	1
A6				VOLTAGE REGULATOR	NAT.	7805	N.I.V.	1
A7				TRANSISTOR		BD681	N.I.V.	1
C1,17	2.2MFD	20		CAPACITOR TANT.	KEMET		BCK-3225	2
C2	0.56MFD			CAPACITOR CERAMIC	VIT.		BCA-5564	1
C3,10	10MFD	20		CAPACITOR TANT.	KEMET		BCK-3106	2
C4,7	0.39MFD			CAPACITOR CERAMIC	VIT.		BCA-5394	2
C5,19	1MFD			CAPACITOR CERAMIC	VIT.		BCA-5105	2
C8,9	1000 MFD	25		CAPACITOR ELECTRO.	PHILIPS		BCE-3108	2
C11	5.6MFD	10		CAPACITOR TANT.	SPRAGUE		BCJ-2565	1
C12	22PF			CAPACITOR CERAMIC	VIT.		BCA-1220	1
C13	5 - 60PF			TRIMMER	PHILIPS		BCY-31	1
C14, 15,16,18	0.1MFD			CAPACITOR CERAMIC	VIT.		BCA-1104	4
C6,10				NOT USED				
CR1,2				DIODE		IN914	BVD-1N914	2
CR3	8.2V			ZENER DIODE			BVD-3082	1
CR4				DIODE		IN4004	BVD-IN4004	1
DS1				L.C.DISPLAY	OPTREX	MC16-106A	N.I.V.	1
R1	1K8		2	RESISTOR 0.5W			BRA-55182	1
R2	8K3		2	RESISTOR 0.5W			BRA-55822	1
R3	160K		2	RESISTOR 0.5W			BRA-55164	1
R4	82K		2	RESISTOR 0.5W			BRA-55823	1
R5	39K		2	RESISTOR 0.5W			BRA-55393	1
R6	20K		2	RESISTOR 0.5W			BRA-55203	1
R7,8 17,19,27,34	10K		2	RESISTOR 0.5W			BRA-55103	6
R9,10 13,16,21,22,23,24,25,26,42,43,44,45	100K		2	RESISTOR 0.5W			BRA-55104	14
R11, 20,35,41	51K		2	RESISTOR 0.5W			BRA-55513	4
R12	10K			S.I.L. RESISTOR PACK	A.B.	110A103	N.I.V.	1
R14, 15	6K2		2	RESISTOR 0.5W			BRA-55622	2
R18	50K	LIN		POTENTIOMETER 0.5W	I.R.C.		BPA-1503	1
R28, 29,30,31	4K3		2	RESISTOR 0.5W			BRA-55432	4
R32, 33	10K	LIN		TRIM POT.	A.B.	85P103	N.I.V.	2

ITEM	VALUE	Vw	%	TYPE	MAKE	REF	BMR VOCAB	QTY
R46	330K		2	RESISTOR 0.5W			BRA-55334	1
R47	1K		2	RESISTOR 0.5W			BRA-55102	1
R35,36,37,38,39,40,41				NOT USED				
S1,2	S.P.S.T			PUSH BUTTON SWITCH MOM.	C&K	8221	BSP-8	2
S3	D.P.D.T.			SWITCH TOGGLE	C&K	7201	BST-23	1
S4	7 POLE			SWITCH ROTARY	N.S.F.		BSW-101	1
							BSW-114	1
							BSW-134	2
XTAL	32.768KHz			CRYSTAL			N.I.V,	1
J1				CONNECTOR	CANNON	KPT02E 14-19-S	BKF-025	1
				PLUG	CANNON	KPT06F 14-19-P	BKF-126	2
				DIECAST BOX	E-STONE	6827/P	BHB-8	1

Drawn	Date	Checked	Amendments	Issue
[illegible]	12/5/86			1
[illegible]	26/5/86		VARIOUS	2

R8704504





Drawing No.	Sheet No.	Title:
S18	1	Accelerograph Timing System

D

```
        OPT    PAG
*
* PROGRAM TO WRITE TIME TO STRONG MOTION DETECTOR FILM
*
* ON POWER UP PROGRAM WILL READ STATUS OF PD2
*
* ON A '0'
*  PROGRAM WILL READ THE CLOCK & WRITE TIME TO THE FILM LED
*
* ON A '1'
*  THE PROGRAM WILL READ THE CLOCK
*  AND DISPLAY THE TIME ON A L.C.D. DISPLAY
*  FOR CHECKING AND SETTING UNDER CONTROL OF A
*  ONE SECOND INTERRUPT FROM THE CLOCK
*
* DATA FOR L.C.D. ON BITS 0-7 PORT C
* ADDRESS FOR CLOCK ON BITS 4-7 PORT C
* DATA TO/FROM THE CLOCK ON BITS 0-3 PORT C
* CONTROL FOR L.C.D. ON BITS 0-2 PORT A
* CONTROL FOR CLOCK ON BITS 4-7 PORT A
* DATA TO L.E.D. ON BITS 0-3 PORT B
* MODE CONTROL ON BIT 2 PORT D
* DIGIT CONTROL ON BIT 0 PORT D
* ADVANCE CONTROL ON BIT 1 PORT D
* 1HZ. INTERRUPT ON BIT 1 PORT C
*
```

```
                *
                *  EQUATES FOLLOW
                *
0000  PAD       EQU     $00         PORT A DATA
0001  PBD       EQU     $01         PORT B DATA
0002  PCD       EQU     $02         PORT C DATA
0003  PDD       EQU     $03         PORT D DATA
0004  PAC       EQU     $04         PORT A CONTROL
0005  PBC       EQU     $05         PORT B CONTROL
0006  PCC       EQU     $06         PORT C CONTROL
0006  HOLD      EQU     6           HOLD BIT
0004  RD        EQU     4           CLOCK READ BIT
0005  WR        EQU     5           CLOCK WRITE BIT
0007  CS        EQU     7           CLOCK CHIP SELECT BIT
0000  DIGIT     EQU     0           DIGIT SELECT BIT
0001  ADV       EQU     1           DIGIT ADVANCE BIT
0002  MODE      EQU     2           NORMAL/SET SELECT
0003  READY     EQU     3           READY TO WRITE TO FILM
00FF  PCWR      EQU     $FF         PORT C WRITE CONTROL BYTE
00F0  PCRD      EQU     $F0         PORT C READ CONTROL BYTE
0009  DAY1M     EQU     9           MAX VALUE DAY *1
0003  DAY10M    EQU     3
0009  MO1M      EQU     9
0001  MO10M     EQU     1
0009  YR1M      EQU     9
0009  YR10M     EQU     9
0009  MIN1M     EQU     9
0005  MIN10M    EQU     5
0009  HR1M      EQU     9
0002  HR10M     EQU     2
0020  UPDATE    EQU     $20
0010  YR10      EQU     $010
0011  YR1       EQU     $011
                * $012 BLANK FOR COLON
0013  MO10      EQU     $013
0014  MO1       EQU     $014
                * $015 BLANK FOR COLON
0016  DAY10     EQU     $016
0017  DAY1      EQU     $017
                * $018 BLANK FOR COLON
0019  HR10      EQU     $019
001A  HR1       EQU     $01A
                * $01B BLANK FOR COLON
001C  MIN10     EQU     $01C
001D  MIN1      EQU     $01D
001E  SEC10     EQU     $01E
001F  SEC1      EQU     $01F
0002  REGS      EQU     $02         L.C.D. REGISTER SELECT
0001  RNW       EQU     $01         L.C.D. READ/WRITE LINE
0000  EN        EQU     $00         L.C.D. ENABLE LINE
```

```
                        *
                        *
0F38                            ORG     $F38        MASK OPTION REGISTER
0F38 80                         FCB     $80         RC OSCILLATOR
                        *
0FFA                            ORG     $FFA        VECTORS ALL TRAPPED BACK TO START
0FFA 0100                       FDB     PROG        INTRUPT
0FFC 0100                       FDB     PROG        SWI
0FFE 0100                       FDB     PROG        RESET
                        *
                        *
                        *  MAIN PROGRAM FOLLOWS
0100                            ORG     $100        START OF MAIN EPROM
0100 A6 40              PROG    LDA     #%01000000 NO INT. 2
0102 B7 0A                      STA     $0A         MISC REGISTER
0104 9B                         SEI                 NO INT. 1
0105 A6 00                      LDA     #$00        PRESET PORT VALUES
0107 B7 00                      STA     PAD
0109 B7 02                      STA     PCD
010B A6 FF                      LDA     #$FF        LEDS OFF
010D B7 01                      STA     PBD
010F A6 FF                      LDA     #$FF
0111 B7 04                      STA     PAC         PORT A OUTPUT
0113 B7 05                      STA     PBC         PORT B OUTPUT
0115 A6 F0                      LDA     #PCRD
0117 B7 06                      STA     PCC         PORT C READ MODE
                        *
                        * READ BIT 2 PORT D AND BRANCH IF NECESSARY
                        *
0119 04 03 03                   BRSET   MODE,PDD,LCDON
011C CC 0345                    JMP     FILM        GO WRITE DATA TO FILM
                        *
                        * SET COLONS IN RAM AREA
                        *
011F A6 0A              LCDON   LDA     #$0A        ASCII FOR COLON AFTER
0121 B7 12                      STA     $12         $30 ADDED LATER
0123 B7 15                      STA     $15
0125 B7 18                      STA     $18
0127 B7 1B                      STA     $1B
                        *
                        * PRESET UPDATE CODE FOR NO UPDATE
                        *
0129 A6 FF                      LDA     #$FF
012B B7 20                      STA     UPDATE
```

```
                        *
                        *PROGRAM L.C.D. TO ACCEPT DATA
                        *
012D A6 FF                      LDA     #PCWR       PORT C WRITE MODE
012F B7 06                      STA     PCC
0131 15 00                      BCLR    REGS,PAD REGISTER 0
0133 13 00                      BCLR    RNW,PAD WRITE MODE
0135 A6 30                      LDA     #%00110000 FUNCTION SET FOR 8 BITS
0137 AD 16                      BSR     LCDWR       GO SEND IT
0139 A6 30                      LDA     #%00110000 REPEAT COMMAND
013B AD 12                      BSR     LCDWR
013D A6 34                      LDA     #%00110100 8 BITS 1 LINE 5*10 DOTS
013F AD 0E                      BSR     LCDWR
0141 A6 06                      LDA     #%00000110 CURSOR INCREMENT NO SHIFT
0143 AD 0A                      BSR     LCDWR
0145 A6 0C                      LDA     #%00001100 DISPLAY ON CURSOR AND BLINK OFF
0147 AD 06                      BSR     LCDWR
0149 A6 F0                      LDA     #PCRD       CHANGE PORT C TO READ MODE
014B B7 06                      STA     PCC
014D 20 0C                      BRA     TIME
                        *
                        * ROUTINE TO WRITE DATA TO L.C.D. & DELAY 40 MICRO SECOND
                        * BEFORE RETURNING
                        * ENTRY WITH DATA IN A
                        *
014F B7 02              LCDWR   STA     PCD
0151 10 00                      BSET    EN,PAD TOGGLE ENABLE LINE
0153 11 00                      BCLR    EN,PAD
0155 A6 05                      LDA     #$05        40 MICROSEC DELAY
0157 4A                 LOOP1   DECA
0158 26 FD                      BNE     LOOP1
015A 81                         RTS                 RETURN
                        *
                        * ROUTINE TO DISPLAY TIME FOR UPDATING
                        *
                        * SET UP 1 HZ INTERRUPT PULSE FOR BIT 1 PORT C
                        *
015B A6 F0              TIME    LDA     #PCRD       JUST IN CASE
015D B7 06                      STA     PCC
015F A6 F0                      LDA     #$F0        SET ALL ADDRESS LINE HIGH
0161 B7 02                      STA     PCD
0163 18 00                      BSET    RD,PAD SET READ LINE
0165 1E 00                      BSET    CS,PAD SET CHIP SELECT
                        *
                        * POLE BIT 1 PORT C FOR INTERRUPT
                        *
0167 9D                 DIZZY   NOP
0168 02 02 FC                   BRSET   1,PCD,DIZZY NO PULSE TRY AGAIN
```

```
                        *
                        * HAS RECIEVED AN INTERRUPT TURN OFF 1 HZ PULSE
                        *
016B 19 00                      BCLR    RD,PAD
016D 1F 00                      BCLR    CS,PAD
                        *
                        * TEST FOR DIGIT ADVANCE MODES
                        *
016F 01 03 06                   BRCLR   DIGIT,PDD,ADIGIT GO ADVANCE DIGIT
0172 03 03 34                   BRCLR   ADV,PDD,ADVA GO ADVANCE DATA
0175 CC 02F5                    JMP     NTIME
                        *
                        * ADIGIT ROUTINE ADVANCES THE UPDATE REGISTER BY THE
                        * FOLLOWING MODE FF,00,01,03,04,06,07,09,0A,0C,0D
                        *   & REPEAT
                        *
0178 B6 20              ADIGIT  LDA     UPDATE
017A A1 FF                      CMP     #$FF            CHECK FOR NO UPDATE
017C 27 24                      BEQ     UFF             YES GO START
017E A1 01                      CMP     #$01
0180 27 12                      BEQ     UTWO            UPDATE TWICE
0182 A1 04                      CMP     #$04
0184 27 0E                      BEQ     UTWO
0186 A1 07                      CMP     #$07
0188 27 0A                      BEQ     UTWO
018A A1 0A                      CMP     #$0A
018C 27 06                      BEQ     UTWO
018E A1 0D                      CMP     #$0D
0190 27 09                      BEQ     UOD             GO RESET
0192 20 02                      BRA     UONE            UPDATE ONCE ONLY
                        *
0194 3C 20              UTWO    INC     UPDATE
0196 3C 20              UONE    INC     UPDATE
0198 CC 02F5                    JMP     NTIME           GO READ TIME
                        *
019B A6 FF              UOD     LDA     #$FF            RESET UPDATE
019D B7 20                      STA     UPDATE
019F CC 02F5                    JMP     NTIME
                        *
01A2 A6 00              UFF     LDA     #$00            START AGAIN
01A4 B7 20                      STA     UPDATE
01A6 CC 02F5                    JMP     NTIME
```

```
                        *
                        *ADVA ROUTINE WILL ADVANCE DATA THAT UPDATE POINTS TO
                        *
01A9 B6 20      ADVA    LDA     UPDATE
01AB A1 FF              CMP     #$FF        CHECK FOR NO UPDATE
01AD 26 03              BNE     ADV1        NOT A BYPASS CODE, CHECK REST
01AF CC 02F5            JMP     NTIME
01B2 A1 00      ADV1    CMP     #$00        CHECK FOR DATA TO UPATE
01B4 27 5F              BEQ     Y10
01B6 A1 01              CMP     #$01
01B8 27 52              BEQ     Y1
01BA A1 03              CMP     #$03
01BC 27 43              BEQ     M10
01BE A1 04              CMP     #$04
01C0 27 35              BEQ     M1
01C2 A1 06              CMP     #$06
01C4 27 25              BEQ     D10
01C6 A1 07              CMP     #$07
01C8 27 17              BEQ     D1
01CA A1 09              CMP     #$09
01CC 27 6D              BEQ     H10
01CE A1 0A              CMP     #$0A
01D0 27 60              BEQ     H1
01D2 A1 0C              CMP     #$0C
01D4 27 51              BEQ     MI10
01D6 A1 0D              CMP     #$0D
01D8 27 44              BEQ     MI1
                        *
                        * IF CODE GETS HERE RESET UPDATE TO $FF
                        *
01DA A6 FF              LDA     #$FF
01DC B7 20              STA     UPDATE
01DE CC 02F5            JMP     NTIME
```

```
                        *
                        * MODIFY ROUTINES FOLLOW
                        *
 01E1 AE 70        D1       LDX     #$70        CLOCK ADDRESS OF DAY *1
 01E3 CD 026E               JSR     SREAD       READ DATA INTO A
 01E6 4C                    INCA                INCREMENT DATA
 01E7 A1 09                 CMP     #DAY1M      CHECK FOR MAX VALUE
 01E9 20 61                 BRA     WRITE
 01EB AE 80        D10      LDX     #$80        DAY *10
>01ED CD 026E               JSR     SREAD
 01F0 A4 03                 AND     #$03        MASK FOR BITS 0-1
 01F2 4C                    INCA
 01F3 A1 03                 CMP     #DAY10M
 01F5 20 55                 BRA     WRITE
 01F7 AE 90        M1       LDX     #$90        MONTH *1
>01F9 CD 026E               JSR     SREAD
 01FC 4C                    INCA
 01FD A1 09                 CMP     #MO1M
 01FF 20 4B                 BRA     WRITE
 0201 AE A0        M10      LDX     #$A0        MONTH *10
 0203 AD 69                 BSR     SREAD
 0205 A4 01                 AND     #$01        MASK FOR BIT 0 ONLY
 0207 4C                    INCA
 0208 A1 01                 CMP     #MO10M
 020A 20 40                 BRA     WRITE
 020C AE B0        Y1       LDX     #$B0        YEAR *10
 020E AD 5E                 BSR     SREAD
 0210 4C                    INCA
 0211 A1 09                 CMP     #YR1M
 0213 20 37                 BRA     WRITE
 0215 AE C0        Y10      LDX     #$C0        YEAR *10
 0217 AD 55                 BSR     SREAD
 0219 4C                    INCA
 021A A1 09                 CMP     #YR10M
 021C 20 2E                 BRA     WRITE
 021E AE 20        MI1      LDX     #$20        MINUTES *1
 0220 AD 4C                 BSR     SREAD
 0222 4C                    INCA
 0223 A1 09                 CMP     #MIN1M
 0225 20 25                 BRA     WRITE
 0227 AE 30        MI10     LDX     #$30        MINUTES *10
 0229 AD 43                 BSR     SREAD
 022B A4 07                 AND     #$07        MASK FOR BIT 0-2
 022D 4C                    INCA
 022E A1 05                 CMP     #MIN10M
 0230 20 1A                 BRA     WRITE
 0232 AE 40        H1       LDX     #$40        HOUR *1
 0234 AD 38                 BSR     SREAD
 0236 4C                    INCA
 0237 A1 09                 CMP     #HR1M
 0239 20 11                 BRA     WRITE
 023B AE 50        H10      LDX     #$50        HOUR *10
```

```
023D AD 2F              BSR     SREAD
023F A4 03              AND     #$03        MASK FOR BIT 0-1 ONLY
0241 4C                 INCA
0242 A1 02              CMP     #HR10M
0244 23 02              BLS     WHR10       BYPASS CLEAR CODE
0246 A6 00              LDA     #0          CLEAR TIME
0248 AA 08        WHR10 ORA     #$08        SET 24 HOUR MODE
024A A1 FF              CMP     #$FF        FORCE FLAGS TO SUIT BLS
```

```
                        *
                        * WRITE CODE CHECKS FOR BLS FLAGS
                        * CLEARS A IF NECESSARY AND WRITES TIME TO CLOCK
                        *
024C 23 02              WRITE   BLS     SWRITE      BYPASS ZERO CODE
024E A6 00                      LDA     #$00        ZERO A REG
0250 BF 7F              SWRITE  STX     $7F         SAVE IN TEMP RAM
0252 BA 7F                      ORA     $7F         OR RAM WITH DATA
0254 AE FF                      LDX     #PCWR       WRITE CODE
0256 BF 06                      STX     PCC         BITS 0-7 OUTPUT
0258 B7 02                      STA     PCD         WRITE ADDRESS AND DATA
025A 9D                         NOP
025B 9D                         NOP                 DELAY FOR SET UP TIME
025C 1E 00                      BSET    CS,PAD SET CHIP SELECT
025E 1A 00                      BSET    WR,PAD SET WRITE LINE
0260 1B 00                      BCLR    WR,PAD CLEAR WRITE LINE
0262 3F 00                      CLR     PAD         CLEAR HOLD AND CHIP SELECT
0264 AE F0                      LDX     #PCRD       READ CODE
0266 BF 06                      STX     PCC         RESET PORT A I/O
0268 CD 037F                    JSR     CLEAP       CHECK LEAP YEAR
026B CC 02F5                    JMP     NTIME
                        *
                        * SREAD WILL READ THE ADDRESS CONTAINED IN X
                        *  AND RETURN WITH DATA IN A AND ADDRESS IN X
                        *  HOLD LINE WILL BE SET AND REMAIN SET
                        *
026E 3F 00              SREAD   CLR     PAD         START AFRESH
0270 1C 00                      BSET    HOLD,PAD SET HOLD LINE
0272 AD 7B                      BSR     DEL200      DELAY FOR SET UP TIME
0274 BF 02                      STX     PCD         STORE ADDRESS
0276 9D                         NOP
0277 9D                         NOP                 SET UP TIME
0278 1E 00                      BSET    CS,PAD SET CHIP SELECT
027A 18 00                      BSET    RD,PAD SET READ LINE
027C B6 02                      LDA     PCD         GET DATA IN A
027E 19 00                      BCLR    RD,PAD RESET READ LINE
0280 1F 00                      BCLR    CS,PAD RESRT CHIP SELECT
0282 A4 0F                      AND     #$0F        MASK OFF ADDRESS
0284 81                         RTS                 RETURN TO CALLER
```

```
                        *
                        * CREAD ROUTINE READ CLOCK INTO RAM
                        *
0285 1C 00      CREAD   BSET    HOLD,PAD SET HOLD LINE
0287 1E 00              BSET    CS,PAD JUST IN CASE
0289 18 00              BSET    RD,PAD DITTO
028B AD 62              BSR     DEL200     DELAY FOR SET UP TIME
028D A6 C0              LDA     #%11000000 YEAR *10 ADDRESS
028F AD 55              BSR     READ       READ THE DATA
0291 B7 10              STA     YR10       SAVE DATA IN RAM
0293 A6 B0              LDA     #%10110000 YEAR *1
0295 AD 4F              BSR     READ
0297 B7 11              STA     YR1
0299 A6 A0              LDA     #%10100000 MONTH *10
029B AD 49              BSR     READ
029D A4 01              AND     #$01       MASK FOR BIT 0 ONLY
029F B7 13              STA     MO10
02A1 A6 90              LDA     #%10010000 MONTH *1
02A3 AD 41              BSR     READ
02A5 B7 14              STA     MO1
02A7 A6 80              LDA     #%10000000 DAY *10
02A9 AD 3B              BSR     READ
02AB A4 03              AND     #$03       MASK FOR BITS 0 & 1 ONLY
02AD B7 16              STA     DAY10
02AF A6 70              LDA     #%01110000 DAY *1
02B1 AD 33              BSR     READ
02B3 B7 17              STA     DAY1
02B5 A6 50              LDA     #%01010000 HOUR *10
02B7 AD 2D              BSR     READ
02B9 A4 03              AND     #$03
02BB B7 19              STA     HR10
02BD A6 40              LDA     #%01000000 HOUR *1
02BF AD 25              BSR     READ
02C1 B7 1A              STA     HR1
02C3 A6 30              LDA     #%00110000 MINUTES *10
02C5 AD 1F              BSR     READ
02C7 A4 07              AND     #$07       MASK FOR BIT 0-2 ONLY
02C9 B7 1C              STA     MIN10
02CB A6 20              LDA     #%00100000 MINUTES *1
02CD AD 17              BSR     READ
02CF B7 1D              STA     MIN1
02D1 A6 10              LDA     #%00010000 SECONDS *10
02D3 AD 11              BSR     READ
02D5 A4 07              AND     #$07
02D7 B7 1E              STA     SEC10
02D9 A6 00              LDA     #$00       SECONDS *1
02DB AD 09              BSR     READ
02DD B7 1F              STA     SEC1
02DF 19 00              BCLR    RD,PAD RESET READ LINE
02E1 1F 00              BCLR    CS,PAD
02E3 1D 00              BCLR    HOLD,PAD
02E5 81                 RTS
```

```
                        *
                        *SUBROUTINE READ
                        *
                        * CALLED WITH ADDRESS IN REG A BITS 4-7
                        *   ROUTINE:
                        *       1. SETS ADDRESS LINES
                        *       2. DELAYS FOR SETUP TIME OF 4 MICRO SECONDS
                        *       3. READS TIME BITS INTO REG A
                        *       4. MASKS OF TOP FOUR BITS NOT USED
                        *       5. RETURN WITH DATA IN REG A BITS 0-3
                        *
02E6 B7 02              READ    STA     PCD         SET ADDRESS LINES
02E8 9D                         NOP
02E9 9D                         NOP                 DELAY
02EA B6 02                      LDA     PCD         GET DATA
02EC A4 0F                      AND     #$0F        MASK OFF TOP 4 BITS
02EE 81                         RTS                 RETURN TO CALLER
                        *
                        *DEL200 ROUTINE DELAY FOR APPROX
                        * 200 MICRO SECONDS
                        * THEN RETURNS. USES A REG ONLY
                        *
02EF A6 12              DEL200  LDA     #$12
02F1 4A                 LOOP    DECA
02F2 26 FD                      BNE     LOOP
02F4 81                         RTS
                        *
                        * NTIME ROUTINE WILL READ DATA FROM CLOCK INTO RAM
                        * THEN WRITE THE DATA TO L.C.D. AND POSITION THE CURSOR
                        * AS PER THE DATA IN THE UPDATE REGISTER
                        *
02F5 AD 8E              NTIME   BSR     CREAD       GET TIME INTO RAM
02F7 A6 FF                      LDA     #PCWR       CHANGE TO WRITE MODE
02F9 B7 06                      STA     PCC
02FB AE 10                      LDX     #YR10       GET ADDRESS OF DATA IN RAM
02FD 15 00                      BCLR    REGS,PAD
02FF 13 00                      BCLR    RNW,PAD
0301 A6 0C                      LDA     #%00001100 CURSOR OFF
0303 CD 014F                    JSR     LCDWR
0306 A6 80                      LDA     #$80        CURSOR ADDRESS ZERO
0308 CD 014F                    JSR     LCDWR
030B 14 00                      BSET    REGS,PAD REG1 FOR DATA
030D F6                 LOOP2   LDA     ,X          GET DATA FROM RAM
030E AB 30                      ADD     #$30        CONVERT DATA TO ASCII
0310 CD 014F                    JSR     LCDWR       SEND TO L.C.D.
0313 5C                         INCX                POINT TO NEXT DATA
0314 A3 20                      CPX     #$20        CHECK FOR FINISHED
0316 26 F5                      BNE     LOOP2       NO KEEP GOING
0318 15 00                      BCLR    REGS,PAD RESET TO REG 0
```

```
                        *
                        * CHECK IF CURSOR SHOULD BE ON
                        *
031A B6 20              UPDATE  LDA     UPDATE
031C A1 FF                      CMP     #$FF
031E 27 0A                      BEQ     FTIME       NO SO FINISHED
                        *
                        * TURN CURSOR ON AND POSITION ON UPDATE CHAR
                        *
0320 AB 80                      ADD     #$80        CONVERT UPDATE CODE TO CURSOR ADDR
0322 CD 014F                    JSR     LCDWR       SEND IT
0325 A6 0E                      LDA     #%00001110 DISPLAY ON, CURSOR ON, BLINK OFF
0327 CD 014F                    JSR     LCDWR
                        *
032A A6 F0              FTIME   LDA     #PCRD
032C B7 06                      STA     PCC
                        *
                        * FINISHED NO GO WAIT FOR NEXT INT.
032E CC 015B                    JMP     TIME
                        *
                        *FWRITE ROUTINE IS CALLED WITH DATA IN REG A
                        * COMPLEMENTS DATA SINCE LEDS ARE NEGATIVE LOGIC
                        * WRITES DATA TO LEDS - PORT B BITS 0-3
                        * DELAYS 0.25 SEC, CLEARS LEDS AND RETURNS
                        *
0331 43                 FWRITE  COMA                FOR NEGATIVE LOGIC
0332 B7 01                      STA     PBD         WRITE DATA TO LEDS
                        *
                        * DELAY FOR 0.25 SECONDS
                        *
0334 AE 28                      LDX     #40
0336 5A                 LOOP3   DECX                OUTER LOOP
0337 27 07                      BEQ     TIMOUT      WHEN FINISHED
0339 A6 FF                      LDA     #$FF
033B 4A                 LOOP4   DECA                INNER LOOP
033C 27 F8                      BEQ     LOOP3       WHEN ZERO
033E 20 FB                      BRA     LOOP4       KEEP GOING
                        *
0340 A6 FF              TIMOUT  LDA     #$FF
0342 B7 01                      STA     PBD         TURN LEDS OFF
0344 81                         RTS
```

```
                        *
                        * FILM ROUTINE WILL READ TIME FROM CLOCK AND WRITE
                        * TIME OUT TO FILM LEDS USING A 0.25 SEC DELAY
                        * AFTER CHECKING STATUS OF READY BIT
                        *
0345 07 03 02           FILM    BRCLR   READY,PDD,READY0
0348 20 FB                      BRA     FILM        LOOP HERE TILL READY = 0
                        *
034A 06 03 02           READY0  BRSET   READY,PDD,READY1
034D 20 FB                      BRA     READY0      LOOP HERE TILL READY =1
                        *
034F CD 0285            READY1  JSR     CREAD
0352 B6 10                      LDA     YR10
0354 AD DB                      BSR     FWRITE
0356 B6 11                      LDA     YR1
0358 AD D7                      BSR     FWRITE
035A B6 13                      LDA     MO10
035C AD D3                      BSR     FWRITE
035E B6 14                      LDA     MO1
0360 AD CF                      BSR     FWRITE
0362 B6 16                      LDA     DAY10
0364 AD CB                      BSR     FWRITE
0366 B6 17                      LDA     DAY1
0368 AD C7                      BSR     FWRITE
036A B6 19                      LDA     HR10
036C AD C3                      BSR     FWRITE
036E B6 1A                      LDA     HR1
0370 AD BF                      BSR     FWRITE
0372 B6 1C                      LDA     MIN10
0374 AD BB                      BSR     FWRITE
0376 B6 1D                      LDA     MIN1
0378 AD B7                      BSR     FWRITE
                        *
037A AD 03                      BSR     CLEAP       CHECK LEAP YEAR
037C 9D                 IDLE    NOP
037D 20 FD                      BRA     IDLE        LOOP HERE TILL POWER OFF
```

```
                        *
                        * CHECK FOR IMPENDING LEAP YEAR
                        *  IF BIT 0&1 OF YEAR = 0 THEN THIS IS A LEAP YEAR
                        *  IF BIT 0&1 OF YEAR = 1 THEN THIS IS YEAR PRIOR TO LEAP
                        * LEAP YEAR FLAG IS RESET BY CLOCK
                        *
037F 1C 00              CLEAP   BSET    HOLD,PAD
0381 1E 00                      BSET    CS,PAD
0383 18 00                      BSET    RD,PAD READ TIME FROM CLOCK BUT HOLD HOLD
0385 CD 02EF                    JSR     DEL200      DELAY
0388 A6 C0                      LDA     #$C0        YEAR 10
038A CD 02E6                    JSR     READ
038D B7 10                      STA     YR10
038F A6 B0                      LDA     #$B0        YEAR *1
0391 CD 02E6                    JSR     READ
0394 B7 11                      STA     YR1
0396 A6 A0                      LDA     #$A0        MONTH *10
0398 CD 02E6                    JSR     READ
039B A4 01                      AND     #$01        MASK FOR BIT 0 ONLY
039D B7 13                      STA     MO10
039F A6 90                      LDA     #$90        MONTH *1
03A1 CD 02E6                    JSR     READ
03A4 B7 14                      STA     MO1
03A6 A6 80                      LDA     #$80        DAY *10
03A8 CD 02E6                    JSR     READ
03AB A4 03                      AND     #$03        MASK FOR BITS 0 & 1 ONLY
03AD B7 16                      STA     DAY10
03AF 19 00                      BCLR    RD,PAD RESET READ
03B1 1F 00                      BCLR    CS,PAD RESET CHIP SELECT
                        * LEAVE HOLD LINE HIGH
03B3 4F                         CLRA                ZERO ACCUMULATOR
03B4 01 10 02                   BRCLR   0,YR10,TEST10 CONVERT BCD TO BINARY
03B7 AB 0A                      ADD     #10
03B9 03 10 02           TEST10  BRCLR   1,YR10,TEST20
03BC AB 14                      ADD     #20
03BE 05 10 02           TEST20  BRCLR   2,YR10,TEST30
03C1 AB 28                      ADD     #40
03C3 07 10 02           TEST30  BRCLR   3,YR10,TEST0
03C6 AB 50                      ADD     #80
03C8 01 11 02           TEST0   BRCLR   0,YR1,TEST1
03CB AB 01                      ADD     #1
03CD 03 11 02           TEST1   BRCLR   1,YR1,TEST2
03D0 AB 02                      ADD     #2
03D2 05 11 02           TEST2   BRCLR   2,YR1,TEST3
03D5 AB 04                      ADD     #4
03D7 07 11 02           TEST3   BRCLR   3,YR1,TEST4
03DA AB 08                      ADD     #8
```

```
                        *
                        * BINARY YEAR NUMBER NOW IN ACC.
                        * MASK FOR BITS 0& 1 ONLY
                        *
03DC A4 03              TEST4   AND     #$03
03DE 27 14                      BEQ     LEAP        YES A LEAP YEAR
03E0 A1 03                      CMP     #$03        CHECK FOR PRIOR TO LEAP
03E2 26 39                      BNE     WAYOUT      NO GO TO EXIT
                        *
                        * THIS IS THE YEAR PRIOR TO A LEAP YEAR
                        * CHECK FOR MONTH 3-12
                        *
03E4 B6 14                      LDA     MO1
03E6 A1 02                      CMP     #$02
03E8 23 02                      BLS     CHMO10      LESS THAN MARCH CHECK FOR > OCT
03EA 20 16                      BRA     SLEAP       SET LEAP YEAR
                        *
                        * CHECK FOR MONTH GREATER THAN 10
                        *
03EC B6 13              CHMO10  LDA     MO10
03EE A1 01                      CMP     #$01        10 OR MORE
03F0 27 10                      BEQ     SLEAP       YES GREATER THAN OCT
03F2 20 29                      BRA     WAYOUT
                        *
                        * LEAP WILL CHECK FOR PRIOR TO MARCH ON A LEAP YEAR
                        *
03F4 B6 14              LEAP    LDA     MO1
03F6 A1 02                      CMP     #$02
03F8 23 02                      BLS     CHMO11      CHECK FOR >OCT
03FA 20 21                      BRA     WAYOUT
                        *
                        * CHECK FOR GREATER THAN OCTOBER
                        *
03FC B6 13              CHMO11  LDA     MO10
03FE A1 01                      CMP     #$01
0400 27 1B                      BEQ     WAYOUT
                        *
                        * SLEAP ROUTINE WILL SET THE LEAP YEAR FLAG
                        *
0402 A6 FF              SLEAP   LDA     #PCWR
0404 B7 06                      STA     PCC
0406 B6 16                      LDA     DAY10
0408 AA 84                      ORA     #%10000100 ADD ADDRESS AND L.Y.FLAG
040A B7 02                      STA     PCD
040C 1E 00                      BSET    CS,PAD
040E 1A 00                      BSET    WR,PAD
0410 9D                         NOP
0411 9D                         NOP
0412 1B 00                      BCLR    WR,PAD
0414 1F 00                      BCLR    CS,PAD
0416 A6 F0                      LDA     #PCRD
0418 B7 06                      STA     PCC
```

```
041A 1D 00              BCLR    HOLD,PAD
041C 81                 RTS                     RETURN TO CALLER
```

```
                        *
                        * WAYOUT WILL RESET LEAP YEAR FLAG, JUST IN CASE
                        *
 041D A6 FF     WAYOUT  LDA     #PCWR
 041F B7 06             STA     PCC
 0421 B6 16             LDA     DAY10
 0423 AA 80             ORA     #%10000000 ADD ADDRESS ONLY
 0425 B7 02             STA     PCD
 0427 1E 00             BSET    CS,PAD
 0429 1A 00             BSET    WR,PAD
 042B 9D                NOP
 042C 9D                NOP
 042D 1B 00             BCLR    WR,PAD
 042F 1F 00             BCLR    CS,PAD
 0431 A6 F0             LDA     #PCRD
 0433 B7 06             STA     PCC
 0435 1D 00             BCLR    HOLD,PAD
 0437 81                RTS                     RETURN TO CALLER

0 ERROR(S) DETECTED
```

SYMBOL TABLE:

ADIGIT	0178	ADV	0001	ADV1	01B2	ADVA	01A9	CHMO10	03EC
CHMO11	03FC	CLEAP	037F	CREAD	0285	CS	0007	D1	01E1
D10	01EB	DAY1	0017	DAY10	0016	DAY10M	0003	DAY1M	0009
DEL200	02EF	DIGIT	0000	DIZZY	0167	EN	0000	FILM	0345
FTIME	032A	FWRITE	0331	H1	0232	H10	023B	HOLD	0006
HR1	001A	HR10	0019	HR10M	0002	HR1M	0009	IDLE	037C
LCDON	011F	LCDWR	014F	LEAP	03F4	LOOP	02F1	LOOP1	0157
LOOP2	030D	LOOP3	0336	LOOP4	033B	M1	01F7	M10	0201
MI1	021E	MI10	0227	MIN1	001D	MIN10	001C	MIN10M	0005
MIN1M	0009	MO1	0014	MO10	0013	MO10M	0001	MO1M	0009
MODE	0002	NTIME	02F5	PAC	0004	PAD	0000	PBC	0005
PBD	0001	PCC	0006	PCD	0002	PCRD	00F0	PCWR	00FF
PDD	0003	PROG	0100	RD	0004	READ	02E6	READY	0003
READY0	034A	READY1	034F	REGS	0002	RNW	0001	SEC1	001F
SEC10	001E	SLEAP	0402	SREAD	026E	SWRITE	0250	TEST0	03C8
TEST1	03CD	TEST10	03B9	TEST2	03D2	TEST20	03BE	TEST3	03D7
TEST30	03C3	TEST4	03DC	TIME	015B	TIMOUT	0340	UOD	019B
UFF	01A2	UONE	0196	UPDATE	0020	UTWO	0194	WAYOUT	041D
WHR10	0248	WR	0005	WRITE	024C	Y1	020C	Y10	0215
YR1	0011	YR10	0010	YR10M	0009	YR1M	0009		